1. Example on Quadratic Fit using Maxima (Posted to Forum by Martin)

$t1 := \text{time}(0)$

$\text{appVersion}(3) = \text{"0.99.7251"}$

$$Data := \begin{bmatrix} 1 & 0.9 \\ 2 & 2.4 \\ 3 & 7 \\ 4 & 17.4 \\ 5 & 30.5 \\ 6 & 48.3 \end{bmatrix}$$

Assumed Quadratic Function

$$f(x) := A + B \cdot x + C \cdot x^2$$

non-linear curve fit, based on Maxima's package lsquares

$$\left[L := \text{Fit}\left(Data, \begin{Bmatrix} x \\ y \end{Bmatrix}, y = f(x), \begin{Bmatrix} A \\ B \\ C \end{Bmatrix}, \begin{Bmatrix} 1 \\ 1 \\ 1 \end{Bmatrix} \right) \right]$$



$$\text{Draw2D}\left(\begin{Bmatrix} points(Data) \\ color = red \\ explicit(f(x), x, 0, 6) \\ grid = true \end{Bmatrix}, \text{"plot.png"} \right)$$

Assignment using boolean equations based on Maxima

$$\text{Assign}(L) = \begin{Bmatrix} 3.83 \\ -4.9604 \\ 2.0625 \end{Bmatrix}$$

$A = 3.83$

$B = -4.9604$

$C = 2.0625$

$X := \text{col}(Data, 1)$

$Y := \text{col}(Data, 2)$

$$Y = \begin{bmatrix} 0.9 \\ 2.4 \\ 7 \\ 17.4 \\ 30.5 \\ 48.3 \end{bmatrix} \qquad Y_{fit} := \overrightarrow{f(X)} = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix} \qquad Y - Y_{fit} = \begin{bmatrix} -0.0321 \\ 0.2407 \\ -0.5114 \\ 0.4114 \\ -0.0907 \\ -0.0179 \end{bmatrix}$$

$\text{MaximaControl}(\text{"cleanup"}) = \text{"Cleanup complete."}$

$\text{time}(0) - t1 = 2.547\ s$

2. Above Example on Quadratic Fit using Native SMath

$t1 := \text{time}(0)$

$x := X$ $y := Y$ → $\text{Clear}(X, Y, A, B, C) = 1$

The data to be fitted $x = \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \end{bmatrix}$ $y = \begin{bmatrix} 0.9 \\ 2.4 \\ 7 \\ 17.4 \\ 30.5 \\ 48.3 \end{bmatrix}$

Just to show what equation we are fitting. We have to find 3x1 matrix "b"

$y_{fit} = b_1 + b_2 \cdot x + b_3 \cdot x^2$

List of functions to make X matrix $\varphi(u) := \begin{bmatrix} 1 \\ u \\ u^2 \end{bmatrix}$

$n := \text{length}(x) = 6$

$k := \text{length}(\varphi(x)) = 3$

```
for j ∈ [1..k]
  for i ∈ [1..n]
    X_{i,j} := φ(x_i)_j
```

$i = 6$

$j = 3$

$k = 3$

$X = \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \\ 1 & 4 & 16 \\ 1 & 5 & 25 \\ 1 & 6 & 36 \end{bmatrix}$ ←

$y = C + Bx + Ax^2$ **Quadratic Fit**

Column 1: Coeffs. of C (All ones)
Column 2: Coeffs. of x
Column 3: Coeffs. of x^2

Compute the two matrices that will be multiplied to give matrix B

$M1 := X^T \cdot X = \begin{bmatrix} 6 & 21 & 91 \\ 21 & 91 & 441 \\ 91 & 441 & 2275 \end{bmatrix}$ $M2 := X^T \cdot y = \begin{bmatrix} 106.5 \\ 538.6 \\ 2853.2 \end{bmatrix}$ $B := M1^{-1} \cdot M2 = \begin{bmatrix} 3.83 \\ -4.9604 \\ 2.0625 \end{bmatrix}$

Alternatively, form the Quadratic Eqn

$F(r) := B_1 + B_2 \cdot r + B_3 \cdot r^2$

$F(x_1) = 0.9321$

$F(0.5) = 1.8654$

$F(0) = 3.83$

$F(8) = 96.1471$

$\overrightarrow{F(x)} = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix}$

Compute the fitted values and plot

```
for i ∈ [1..n]
  yfit_i := B · φ(x_i)
```

value of y when x = 1

$\sum B = 0.9321$

SMath Fit

$yfit = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix}$

Same: compare

Maxima Fit

$Y_{fit} = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix}$ →

$Expt := \text{augment}(x, y)$

$Fit := \text{augment}(x, yfit)$



$\{ Expt, Fit$

X value at min

$x_{min} := \text{solve}\left(\frac{d}{d\,r} F(r) = 0,\ r\right) = 1.2025$

$F(x_{min}) = 0.8476$

$\frac{d^2}{d\,r^2} F(r) = 4.125$

cordinates of minimum point

$Pt\_min := \text{augment}\left(x_{min},\ F(x_{min})\right) = [\,1.2025\ \ 0.8476\,]$



$\{ Expt, F(r), Pt\_min$

$\text{time}(0) - t1 = 0.047\ s$

---

3. Example on Linear Fit using Native SMath

$\text{Clear}\left(x, y, X, B, F(r), n, k, j, Expt, yfit, \varphi(u), x_{min}\right) = 1$

$t1 := \text{time}(0)$

The data to be fitted

$$x := \begin{bmatrix} 0.5 \\ 1.1 \\ 1.5 \\ 2.1 \\ 2.3 \end{bmatrix} \qquad y := \begin{bmatrix} 32 \\ 33 \\ 34.2 \\ 35.1 \\ 35.7 \end{bmatrix}$$

Just to show what equation we are fitting. We have to find 2x1 matrix "b"

$y_{fit} = b_1 + b_2 \cdot x$

List of functions to make X matrix

$$\varphi(u) := \begin{bmatrix} 1 \\ u \end{bmatrix}$$

$n := \text{length}(x) = 5$

$k := \text{length}(\varphi(x)) = 2$

for $j \in [1..k]$
  for $i \in [1..n]$
    $X_{i\,j} := \varphi(x_i)_j$

$i = 5$

$j = 2$

$k = 2$

$$X = \begin{bmatrix} 1 & 0.5 \\ 1 & 1.1 \\ 1 & 1.5 \\ 1 & 2.1 \\ 1 & 2.3 \end{bmatrix}$$

$y = C + Bx$ **Linear fit**

Column 1: Coeffs. of C (All ones)

Column 2: Coeffs. of x

Compute the two matrices that will be multiplied to give matrix B

$$M1 := X^{T} \cdot X = \begin{bmatrix} 5 & 7.5 \\ 7.5 & 13.41 \end{bmatrix} \qquad M2 := X^{T} \cdot y = \begin{bmatrix} 170 \\ 259.42 \end{bmatrix} \qquad B := \text{eval}\left(M1^{-1}\right) \cdot M2 = \begin{bmatrix} 30.9306 \\ 2.0463 \end{bmatrix}$$

Alternatively, form the Quadratic Eqn

$$F(r) := B_1 + B_2 \cdot r$$

$$F(x_1) = 31.9537$$

$$F(0.5) = 31.9537$$

$$F(0) = 30.9306$$

$$F(8) = 47.3009$$

$$\overrightarrow{F(x)} = \begin{bmatrix} 31.9537 \\ 33.1815 \\ 34 \\ 35.2278 \\ 35.637 \end{bmatrix}$$

Compute the fitted values and plot

$$\text{for } i \in [1..n] \quad yfit_i := B \cdot \varphi(x_i)$$

value of y when x = 1

$$\sum B = 32.9769 \quad \rightarrow \quad F(1) = 32.9769$$

$$B = \begin{bmatrix} 30.9306 \\ 2.0463 \end{bmatrix}$$

SMath Fit

$$yfit = \begin{bmatrix} 31.9537 \\ 33.1815 \\ 34 \\ 35.2278 \\ 35.637 \end{bmatrix} \qquad y = \begin{bmatrix} 32 \\ 33 \\ 34.2 \\ 35.1 \\ 35.7 \end{bmatrix} \qquad Expt := \text{augment}(x, y) = \begin{bmatrix} 0.5 & 32 \\ 1.1 & 33 \\ 1.5 & 34.2 \\ 2.1 & 35.1 \\ 2.3 & 35.7 \end{bmatrix}$$



$$\begin{cases} Expt \\ F(r) \end{cases}$$

$$\text{time}(0) - t1 = 0.015\ s$$

$$\text{Clear}(x, y, X, B, F(r), n, k, j, Expt, yfit, \varphi(u), a1, b1, c1, M1, M2) = 1$$

4. Program for Linear & Polynomial Fit using Native SMath

Program for Linear and Polynomial Fit Linear: n = 1 , Poly: n = 2 3, 4, ...

```
Find_Cf(x#, y#, n) := for j ∈ [1..(n + 1)]
                        if j = 1
                          X := x# over x#
                        else
                          X := augment(X, x# over j - 1)
                      M1 := X^T · X
                      M2 := X^T · y#
                      B := eval(M1^-1) · M2
                      [(X · B) B]
```

$t1 := \text{time}(0)$

$$xL := \begin{bmatrix} 0.5 \\ 1.1 \\ 1.5 \\ 2.1 \\ 2.3 \end{bmatrix} \qquad yL := \begin{bmatrix} 32 \\ 33 \\ 34.2 \\ 35.1 \\ 35.7 \end{bmatrix}$$

$$xP := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \end{bmatrix} \qquad yP := \begin{bmatrix} 0.9 \\ 2.4 \\ 7 \\ 17.4 \\ 30.5 \\ 48.3 \end{bmatrix}$$

$$\begin{bmatrix} y1 := Find\_Cf(xL, yL, 1)_1 & B1 := Find\_Cf(xL, yL, 1)_2 \end{bmatrix} = \text{"Linear"}$$
$$\begin{bmatrix} y2 := Find\_Cf(xP, yP, 2)_1 & B2 := Find\_Cf(xP, yP, 2)_2 \end{bmatrix} = \text{"2nd order"}$$
$$\begin{bmatrix} y3 := Find\_Cf(xP, yP, 3)_1 & B3 := Find\_Cf(xP, yP, 3)_2 \end{bmatrix} = \text{"3rd order"}$$
$$\begin{bmatrix} y4 := Find\_Cf(xP, yP, 4)_1 & B4 := Find\_Cf(xP, yP, 4)_2 \end{bmatrix} = \text{"4th order"}$$
$$\begin{bmatrix} y5 := Find\_Cf(xP, yP, 5)_1 & B5 := Find\_Cf(xP, yP, 5)_2 \end{bmatrix} = \text{"5th order"}$$

Linear Fit : x2, y2

$$y1 = \begin{bmatrix} 31.9537 \\ 33.1815 \\ 34 \\ 35.2278 \\ 35.637 \end{bmatrix}$$

Linear Coeffs.

$$B1 = \begin{bmatrix} 30.9306 \\ 2.0463 \end{bmatrix}$$

(2 nd order) : x1, y1

$$y2 = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix}$$

2 nd order: Coeffs

$$B2 = \begin{bmatrix} 3.83 \\ -4.9604 \\ 2.0625 \end{bmatrix}$$

(3 rd order) : x1, y1

$$y3 = \begin{bmatrix} 0.9683 \\ 2.1087 \\ 7.4825 \\ 17.0175 \\ 30.6413 \\ 48.2817 \end{bmatrix}$$

3 rd order: Coeffs

$$B3 = \begin{bmatrix} 4.1333 \\ -5.3419 \\ 2.1889 \\ -0.012 \end{bmatrix}$$

(4 th order) : x1, y1

$$y4 = \begin{bmatrix} 0.9433 \\ 2.1837 \\ 7.4325 \\ 16.9675 \\ 30.7163 \\ 48.2567 \end{bmatrix}$$

4 th order: Coeffs

$$B4 = \begin{bmatrix} 3.0833 \\ -3.5336 \\ 1.216 \\ 0.1921 \\ -0.0146 \end{bmatrix}$$

(5 th order) : x1, y1

$$y5 = \begin{bmatrix} 0.9 \\ 2.4 \\ 7 \\ 17.4 \\ 30.5 \\ 48.3 \end{bmatrix}$$

5 th order: Coeffs

$$B5 = \begin{bmatrix} -16.9 \\ 38.7717 \\ -30.0458 \\ 10.5875 \\ -1.6042 \\ 0.0908 \end{bmatrix}$$

$\text{time}(0) - t1 = 0.047\ s$

EQUATIONS

$$F1(x) := B1_1 + B1_2 \cdot x$$

$$F2(x) := B2_1 + B2_2 \cdot x + B2_3 \cdot x^2$$

$$F3(x) := B3_1 + B3_2 \cdot x + B3_3 \cdot x^2 + B3_4 \cdot x^3$$

$$F4(x) := B4_1 + B4_2 \cdot x + B4_3 \cdot x^2 + B4_4 \cdot x^3 + B4_5 \cdot x^4$$

$$F5(x) := B5_1 + B5_2 \cdot x + B5_3 \cdot x^2 + B5_4 \cdot x^3 + B5_5 \cdot x^4 + B5_6 \cdot x^5$$

$t1 := \text{time}(0)$

$$\overrightarrow{F1(xL)} = \begin{bmatrix} 31.9537 \\ 33.1815 \\ 34 \\ 35.2278 \\ 35.637 \end{bmatrix}$$

$$\overrightarrow{F2(xP)} = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix} \quad \overrightarrow{F3(xP)} = \begin{bmatrix} 0.9683 \\ 2.1087 \\ 7.4825 \\ 17.0175 \\ 30.6413 \\ 48.2817 \end{bmatrix} \quad \overrightarrow{F4(xP)} = \begin{bmatrix} 0.9433 \\ 2.1837 \\ 7.4325 \\ 16.9675 \\ 30.7163 \\ 48.2567 \end{bmatrix} \quad \overrightarrow{F5(xP)} = \begin{bmatrix} 0.9 \\ 2.4 \\ 7 \\ 17.4 \\ 30.5 \\ 48.3 \end{bmatrix}$$

Ex.1: Quadratic Fit (Maxima)

$$Y_{fit} = \begin{bmatrix} 0.9321 \\ 2.1593 \\ 7.5114 \\ 16.9886 \\ 30.5907 \\ 48.3179 \end{bmatrix}$$



$$\begin{cases} \text{augment}(xL, yL) \\ F1(x) \\ \text{augment}(xP, yP) \\ F2(x) \end{cases}$$

X value at function minomum

$$x_{min} := \text{solve}\left(\frac{d}{dx} F2(x) = 0, x\right) = 1.2025$$

Minimum value of F2

$$F2(x_{min}) = 0.8476$$

Alternatively

$$\frac{-B2_2}{2 \cdot B2_3} = 1.2025$$

2nd derivative indicates max or min

$$y'' := \frac{d^2}{dx^2} F2(x) = 4.125$$

Function has a Min.

$\text{time}(0) - t1 = 0.016\ s$